Keywords

^ PURPOSE
This important keyword is the default substitution character.

DESCRIPTION
The ^ signifies that the following parameter name should be replaced with the literal value of that parameter.

SYNTAX
None

EXAMPLES
LET STRING FNAME = CALIB.DAT
READ ^FNAME X1 X2 X3
LOOP FOR K = 1 1 10
   PLOT Y^K
END OF LOOP

NOTE 1
The substitute character can be used effectively with the LOOP command. For example, assume that variables Y1 thru YN and strings S1 and SN have been previously defined:
   LOOP FOR K = 1 1 N
       TITLE Y^K VS X
       LEGEND 1 ^S^K
       PLOT Y^K VS X
   END OF LOOP

Note the distinction between Y^K in the PLOT command and ^S^K in the LEGEND command. For the plot command, we want PLOT Y1 VS X while for the LEGEND command we want to insert the value of the string S1. This is the reason that the Y^K has no preceding ^ while the ^S^K does.

NOTE 2
One interesting application is automatically including (via a LEGEND or X3LABEL command) the literal value of some relevant parameter on a plot. The PROGRAM 2 example demonstrates the placing the fitted coefficients and the residual standard deviation on a post-fit plot.

NOTE 3
The following characters are used to determine the end of the parameter name:
   - a space
   - a left parenthesis
   ) - a right parenthesis
   " - a quote character
   & - an ampersand character
   . - a period
   ^ - another substitute character
   , - a comma
   : - a colon
   ; - a semi-colon
   ] - a right bracket
   + - a plus
   - - a minus
   * - an asterisk
   / - a forward slash

NOTE 4
The substitution is performed before the command is interpreted.
NOTE 5
If the parameter is undefined, it is replaced with a space character in the command line.

NOTE 6
The substitution is done on a parameter name. To replace an element of a variable, do something like the following:

```
LET TEMP=X(22)
LEGEND 1 DATA VALUE = TEMP
```

NOTE 7
The substitution can be nested. This is demonstrated in the program segment in an above NOTE section (TITLE ^(S^K)). Nesting more than 2 deep is rare.

NOTE 8
The VALU() sequence is equivalent to ^ in text strings (e.g., TEXT, LEGEND, TITLE, or LABEL commands). The EXECUTE STRING command can perform some string substitutions. However, the ^ is more general since it can be used in any command (the substitution is performed before the command is passed to routines that process individual commands). For this reason, the VALU() keyword and the EXECUTE STRING command are obsolete.

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS

<table>
<thead>
<tr>
<th>Command</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>SUBSTITUTE CHARACTER</td>
<td>Define the substitute character.</td>
</tr>
<tr>
<td>VALU()</td>
<td>Replace in a text string.</td>
</tr>
<tr>
<td>LOOP</td>
<td>Execute a sequential loop.</td>
</tr>
</tbody>
</table>

APPLICATIONS
Parameter substitution

IMPLEMENTATION DATE
89/12 (early implementations used the “\” character)
PROGRAM 1
LET Y1 = NORMAL RANDOM NUMBER FOR I = 1 1 100
LET Y2 = EXPONENTIAL RANDOM NUMBERS FOR I = 1 1 100
LET NU = 20
LET Y3 = T RANDOM NUMBERS FOR I = 1 1 100
LET Y4 = CAUCHY RANDOM NUMBERS FOR I = 1 1 100
LET STRING T1 = NORMAL RANDOM NUMBERS
LET STRING T2 = EXPONENTIAL RANDOM NUMBERS
LET STRING T3 = T RANDOM NUMBERS
LET STRING T4 = CAUCHY RANDOM NUMBERS
LET X = SEQUENCE 1 1 100
MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
LOOP FOR K = 1 1 4
    TITLE ^T^K
    HISTOGRAM Y^K
END OF LOOP
END OF MULTIPLOT
PROGRAM 2

```plaintext
SKIP 25
READ BERGER1.DAT MEAS TRUE
FIT MEAS TRUE
TITLE AUTOMATIC
CHARACTER X BLANK
LINE BLANK SOLID
X1LABEL INTERCEPT = ^A0   SLOPE = ^A1
X2LABEL RESIDUAL STANDARD DEVIATION = ^RESSD
PLOT MEAS PRED VS TRUE
```

```
INTERCEPT = 4.99368   SLOPE = 0.731112
RESIDUAL STANDARD DEVIATION = 6.080925
```